


Invocaciones entre objetos

GeneXus™

Hasta el momento...


Utilizamos valores fijos para filtrar, en este caso por país



```

1 print Title
2 print ColumnTitles
3 For each Attraction order CountryName
4   where CountryName = 'France'
5   print Attractions
6 endfor

```



```

1 print Title
2 print ColumnTitles
3 For each Attraction order CountryName
4   where CountryId = 2
5   print Attractions
6 endfor

```



¿Y si queremos generalizar el listado y poder “recibir” de alguna forma el país por el cual filtrar?

En situaciones anteriores nos hemos encontrado con la necesidad de llamar a un objeto desde otro.

Por ejemplo, cuando implementamos el objeto procedimiento AttractionsList, necesitamos filtrar las atracciones que tienen como nombre de país “France”, o, lo que era similar, identificador de país igual a 2 (que correspondía a “France”). Y para lograrlo, utilizamos valores fijos en el código.

Pero esto implica que si quisiéramos filtrar las atracciones de un país diferente a Francia, deberíamos modificar el código del procedimiento ¡cada vez!


Object A

Object B
(AttractionsList)

Lo ideal sería que pudiéramos "recibir" de alguna manera en este objeto ese valor por el que queremos filtrar. Dicho de otro modo, que otro objeto GeneXus pueda permitir al usuario elegir ese valor, ... y luego lo envíe a este objeto procedimiento para que liste las atracciones en base a ese país recibido.

Veremos a continuación, mediante este ejemplo, cómo implementar la comunicación entre objetos GeneXus.

Definiendo la comunicación entre objetos.



1) Creamos un web panel que pida el país a considerar.

Variable con tipo de control Dynamic Combo para que ofrezca al usuario los países de la base de datos


2) Agregamos botón para llamar al procedimiento AttractionsList.

Botón que tiene asociado el evento ListAttractiosByCountry

Variable que contiene al país indicado en el form del web panel.

Event:

```
Event 'List attractions by country'
  AttractionsList(&CountryId)
Endevent
```



Para empezar, debemos crear un objeto que sea capaz de ofrecernos una pantalla para pedir valores al usuario y hacer algo con esos valores. Uno de los objetos que permite esto es el web panel, que estudiaremos en detalle más adelante. Por ahora digamos que se trata de un panel visual muy flexible que permite pedir datos al usuario, mostrar información de la base de datos o de otras fuentes, entre otras cosas. Por ejemplo, el Work With de atracciones fue implementado automáticamente por GeneXus como un Web Panel.

Entonces, vamos a crear un objeto de este tipo. Lo llamaremos EnterAttractionsFilter. Veamos que se crea un Form Web, que será la pantalla del objeto. Contiene únicamente una tabla.

Le agregamos una variable, CountryId... la que por nombrar igual que al atributo queda basada en él, y, por lo tanto, asume su mismo tipo de datos. De este modo si le cambiamos el tipo de datos al atributo, por ejemplo, por numérico de 10 en vez de 4, la variable automáticamente cambiará tomando ese nuevo valor.

Ahora editamos las propiedades de la variable y vemos que su propiedad Control Type asume el valor Edit.

Esto significa que cuando se ejecute el web panel este campo esperará que el usuario digite un valor numérico, pero no brindará ninguna ayuda para elegir qué valores existen en la base de datos y a qué países corresponden.

Vamos a cambiar el tipo de control por Dynamic Combo Box. De esta

manera se le va a ofrecer al usuario una serie de valores extraídos de la base de datos, para que él elija el que desea. ¿Qué valores? Los del atributo CountryId. Es decir, se va a recorrer la tabla Country y cargar en el combo los CountryIds existentes. Pero como los identificadores no suelen decirnos nada, si bien la variable va a almacenar un identificador de país, lo que se le va a mostrar al usuario en la pantalla es el contenido del atributo que indiquemos en la propiedad Item Descriptions de la variable.

Elegimos mostrar el nombre de país, CountryName. Vemos cómo aparece la flechita indicadora de combo.

Resumiendo, esto nos ofrecerá en ejecución un combo que nos presentará la lista de países de la base de datos para elegir el que nos interesa.


Agregamos además un botón. Nos pide que indiquemos el nombre del evento que ese botón tendrá asociado. Lo llamamos: "List Attractions By Country". Y vemos que el texto del botón asume el mismo nombre por defecto. Si nos posicionamos sobre él, presionamos botón derecho y elegimos Go to Event.....vemos que se creó un evento con ese nombre, y se pasó de la solapa Web Form a la Events en forma automática, y el cursor está esperando a que ingresemos el código que se ejecutará cuando este evento se dispare. Es decir, cuando el usuario presione el botón asociado.

Lo que necesitamos que se haga en ese momento es llamar al objeto procedimiento AttractionsList que lista las atracciones y enviarle el país que queremos que utilice para filtrarlas:

Notemos que al momento de apretar el botón y ejecutar este código, la variable &CountryId contendrá el identificador de país del país que el usuario haya elegido en el combo box en la pantalla. Anteriormente vimos que una variable es una porción de memoria a la cual le damos un nombre y nos sirve para guardar un dato en forma temporal, y que cada objeto tiene su sección de variables, o sea que las variables que se definen en un objeto son conocidas solamente dentro de ese objeto.

Así, si dos objetos tienen una variable CountryId definida, aunque se llamen igual, se tratará de dos variables distintas.

Definiendo la comunicación entre objetos.



Entonces, ¿cómo hacemos para que un objeto A pueda invocar a otro objeto B en un momento dado, pasándole valores? Y que ese otro objeto B pueda cargar en sus variables internas los valores que se le enviaron, para poder hacer algo con esa información.

Para que un objeto pueda recibir valores (a los que llamamos parámetros), debemos ir a su sección Rules y escribir una regla Parm. Esa regla Parm declara los parámetros que el objeto puede recibir y/o devolver a quien lo llame.

Regla Parm

Para que un objeto pueda recibir valores (parámetros), debemos utilizar la regla Parm.

The image contains three screenshots from the GeneXus IDE:

- Top Left:** Shows the 'Rules' tab for the 'AttractionsList' object. A rule is defined as `Parm(in: &CountryId);`. A callout box points to the `&CountryId` parameter with the text: "Indica que se va a recibir un valor en esa variable."
- Top Right:** Shows the 'Variables' tab for the 'AttractionsList' object. A table lists the variables defined:

Name	Type
&CountryId	Attribute:CountryId

- Bottom:** Shows two side-by-side windows. The left window is 'EnterAttractionsFilter' in the 'Events' tab, showing an event definition: `Event 'List Attractions By Country' AttractionsList(&CountryId) Endevent`. The right window is 'AttractionsList' in the 'Rules' tab, showing the rule `parm(in: &CountryId);` and an `Output_file` statement.

Como en nuestro ejemplo quien va a recibir los valores es el objeto procedimiento AttractionsList, abrimos el objeto y vamos a su sección de reglas. Escribimos...

Observemos que en la Toolbox se nos ofrecen todas las reglas que podríamos escribir en un objeto de este tipo. Entre ellas, la parm. Podríamos haberla arrastrado desde allí.

Vemos también nos informa que tenemos que sustituir esto por un atributo o variable. Luego veremos el caso de los atributos. Por ahora solamente vemos el caso de las variables.

Con "in" estamos indicando que la variable &CountryId será un parámetro de entrada. Esto significa que solamente será utilizado para recibir un valor de quien lo llame. No para devolver. Podemos omitir esta información y dejar que GeneXus la infiera.

Hemos escrito el nombre de la variable, pero no la hemos ingresado como variable en el objeto. Para hacerlo, una de las formas es posicionarnos sobre el nombre, dar botón derecho y elegir "Add variable".

Si ahora vamos a la solapa de variables vemos que se la ha definido, basada, por defecto, en el tipo del atributo CountryId. Esto se debe a que la llamamos igual que a un atributo.

En este objeto hemos definido la variable con el mismo nombre y tipo de datos que la que usamos en el web panel para que el usuario ingresara el país.


Sin embargo, como dijimos, son dos variables distintas. Una válida

solamente en el web panel y la otra en el procedimiento. Podríamos haberlas llamado distinto en ambos objetos, pero para que la comunicación y el pasaje de información sea correcta, el tipo de datos debe ser compatible entre quien llama y quien es llamado.

Ahora, nuestro objeto procedimiento está preparado para recibir a un identificador de país, en este caso desde el webpanel EnterAttractionsFilter.

Regla Parm

Para que un objeto pueda recibir valores (parámetros), debemos utilizar la regla Parm.




```

1 Parm(in: &CountryId);
2

```

Indica que se va a recibir un valor en esa variable.



Name	Type
Variables	
Standard Variables	
CountryId	Attribute:CountryId

Modificamos el Source:

```

print Title
print ColumnTitles
For each Attraction order CountryName
  where CountryId = 2
  print Attractions
endfor

```


Cambiamos a

```

print Title
print ColumnTitles
For each Attraction order CountryName
  where CountryId = &CountryId
  print Attractions
endfor

```

Sólo nos resta quitar el filtro fijo que teníamos (el valor 2 de país) en el For each, y cambiarlo por la variable cuyo valor es recibido como parámetro.




Observemos que al haber definido la regla Parm de esta manera, de ahora en adelante cualquier objeto que llame al procedimiento deberá pasarle el valor del identificador del país.

Ya no podrá invocarse a este procedimiento sin enviarle un valor de este tipo. Es por esta razón que el procedimiento `AttractionsList` ya no va a aparecer en el Developer Menu.

En el caso del webpanel teníamos ese valor en una variable (que el usuario ingresaba en pantalla).

Pero si tuviéramos el dato en un atributo, incluiríamos dentro del paréntesis al atributo que corresponda.


También podríamos pasar directamente un valor.



O en caso de tener que pasar dos o más valores, enviaríamos varios atributos, y/o valores explícitos, y/o variables, separados por coma.

Esos parámetros también se declaran en la regla parm en forma ordenada y separados por comas.

Evidentemente, un objeto que no recibe parámetros no debe declarar regla Parm.



Vamos a probar lo que hemos hecho: F5. Vemos que ya no aparece el procedimiento AttractionsList. Ahora solamente podemos invocarlo a través del web panel.

En el combo del país, elegimos Francia y presionamos el botón.

Al haber elegido el valor France del dynamic combo, internamente se seleccionó el valor del identificador de Francia, en este caso el valor 2, y ese valor es el que se le envía al procedimiento AttractionsList.

Vemos que se ejecuta el reporte, mostrándonos solamente las atracciones del país France.

Listar las atracciones en un rango de nombres determinado

Web panel

The web panel contains a dropdown menu for 'Country Id' with the value '&CountryId'. Below it are two input fields: 'Attraction Name From' with the value '&AttractionNameFrom' and 'Attraction Name To' with the value '&AttractionNameTo'. At the bottom, there are two buttons: 'List Attractions By Country' and 'List Attractions By Name'.

```
Event 'List attractions by name'
  AttractionsByName(&AttractionNameFrom, &AttractionNameTo)
Endevent
```

```
Parm(in:&NameFrom, in:&NameTo);
Output_file('AttractionByName.pdf', 'PDF');
```

```
print Title
print ColumnTitles
For each Attraction order CountryName
  where AttractionName >= &NameFrom
  where AttractionName <= &NameTo
  print Attractions
endfor
```

Proc. AttractionsByName

Name	Type	Is
&Variables		
Standard Variables		
NameFrom	Attribute:AttractionName	
NameTo	Attribute:AttractionName	

Vamos a suponer ahora que queremos listar todas las atracciones cuyos nombres se encuentren entre dos valores elegidos por el usuario. Por ejemplo, entre la "A" y la "D".

Para eso, vamos a agregarle al webpanel que definimos previamente la posibilidad de que el usuario ingrese un nombre inicial y un nombre final para que, presionando un botón, se invoque a un listado que muestre todas las atracciones turísticas cuyos nombres se encuentren dentro de ese rango.

Vamos al webpanel EnterAttractionsFilter y agregamos una tabla con dos variables:

- &AttractionNameFrom... basada en la definición del atributo AttractionName.
- y &AttractionNameTo, también basada en la definición de AttractionName.

Como ya dijimos, esto significa que la definición de la variable está enlazada con la definición del atributo y si cambiamos el tipo de datos del atributo, automáticamente se cambiará la variable en forma acorde. Luego agregamos un botón de evento "List Attractions By Name".

Nos posicionamos en el botón que acabamos de agregar, presionamos botón derecho y elegimos "Go to event". Tenemos que invocar aquí dentro al procedimiento que imprimirá las atracciones turísticas del rango.

Ya teníamos el reporte AttractionsList... pero recibía por parámetro el identificador de país, no el rango de nombres. Vamos a grabarlo con otro nombre, AttractionsByName, y modificar su regla parm, para que ahora reciba dos parámetros de entrada: la variable &NameFrom, y la variable &NameTo.

Tenemos que definir las como variables, y especificarle el tipo de datos, así que hacemos botón derecho sobre la primera y presionamos "Add Variable", y editando sus propiedades, la basamos en el atributo AttractionName.

Hacemos lo mismo con &NameTo.

Ahora vamos a utilizar estas variables/parámetros en el For each, para quedarnos con las atracciones que cumplan que su AttractionName sea mayor o igual al valor de la variable &NameFrom, así como menor o igual al valor de la variable &NameTo.

Con esto el procedimiento está listo y solamente nos resta invocarlo desde el web panel. Vamos al webanel y agregamos la invocación. Arrastramos el reporte AttractionsByName de esta ventana para no tener que digitarlo y entre paréntesis escribimos los parámetros separados por coma, en este caso las variables &AttractionNameFrom y &AttractionNameTo que se le ofrecen al usuario en la pantalla.



Observemos que el primer parámetro que escribimos en la invocación se cargará en el primer parámetro definido en la regla Parm del objeto llamado y el segundo parámetro de la llamada se cargará en el segundo parámetro del objeto invocado.

Debemos ser cuidadosos de respetar el orden en la invocación y en la definición de la regla Parm. Es buena práctica usar nombres relacionados como hicimos aquí, a efectos de entender mejor el código.

Notemos que los nombres que hemos elegido para las variables del web panel y las del procedimiento son distintas. Como ya dijimos, lo importante es que los tipos de datos enviados y recibidos, coincidan.

Presionemos F5 para ejecutar....

En ejecución...



Abrimos el webpanel y para comenzar queremos ver todas las atracciones entre la "A" y la "Z".

Presionamos el botón "List Attractions By Name"... y vemos que se listan todas las atracciones.

Ahora acotamos un poco más el rango, ponemos entre la 'A' y la 'F', y vemos cómo funcionó el filtro.

Listar las atracciones con un único botón y procedimiento.

Web panel

The web panel contains three input fields and a button:

- Country Id:
- Attraction Name From:
- Attraction Name To:
- List Attractions:

```

Event 'List Attractions'
  AttractionsByNameAndCountry(&CountryId, &AttractionNameFrom, &AttractionNameTo)
Endevent
  
```

Proc. AttractionsByNameAndCountry

Name	Type
& Variables	
Standard Variables	
CountryId	Attribute:CountryId
NameFrom	Attribute:AttractionName
NameTo	Attribute:AttractionName

```

Output_file("AttractionsList.pdf", "pdf");
parm(in: &CountryId, in: &NameFrom, in: &NameTo);
  
```

```

print Title
print ColumnTitles
For each Attraction order CountryName
  where AttractionName >= &NameFrom when not &NameFrom.IsEmpty()
  where AttractionName <= &NameTo when not &NameTo.IsEmpty()
  where CountryId = &CountryId when not &CountryId.IsEmpty()
  print Attractions
endfor
  
```

Podríamos también, haber implementado esto mismo con un único procedimiento en lugar de con dos, veamos cómo.

Primero que nada, volvamos al webpanel EnterAttractionsFilter, vamos a configurar la propiedad Empty Item de la variable CountryId en true, y en la propiedad Empty Item Text ingresamos "Select". De esta forma, siempre que ingresemos a este panel, no habrá un país seleccionado por defecto como hasta ahora, sino que aparecerá el texto "Select", y se deberá seleccionar uno.

Ahora creamos un procedimiento, el cual será una mezcla entre AttractionsByName y AttractionsList. Para facilitar su creación, hacemos un "Save As..." de AttractionsByName, y lo llamamos AttractionsByNameAndCountry.

En este nuevo procedimiento, debemos recibir tres parámetros, que corresponderán a los tres filtros que tenemos creados en el webpanel. Por lo que, le sumamos a los que ya tenemos ingresados, la variable CountryId, y la agregamos a la lista de variables de este procedimiento. En la sección Source, agregamos otra cláusula where, para contemplar el filtro por país, where CountryId es igual a la variable CountryId.

Ahora solo nos resta crear el nuevo botón en el web panel para ejecutar este procedimiento. Eliminamos los dos que teníamos creados, y generamos uno de nombre List Attractions.

Hacemos clic derecho sobre él y luego Go To Event, para programar su funcionamiento. Comentamos los eventos anteriores ya que no los vamos a utilizar, y desde el nuevo evento invocamos al nuevo objeto procedimiento creado, `AttractionsByNameAndCountry`, pasándole por parámetro las tres variables que tenemos creadas y que se le ofrecen al usuario en pantalla, `CountryId`, `AttractionNameFrom` y `AttractionNameTo`.

Guardamos los cambios y ejecutamos la aplicación para probar el funcionamiento.

Seleccionamos por ejemplo el país Francia, y queremos que nos muestre solo las atracciones con nombres que comienzan con las letras de la "F" a la "Z". Presionamos el botón List Attractions y vemos que nos filtra correctamente lo que deseamos.

Ahora, si por ejemplo seleccionamos nuevamente el país Francia y no ingresamos filtros por nombre ya que nos interesan todas las atracciones de este país, si presionamos el botón, observamos que no aparece ninguna atracción en el listado.

Lo mismo sucede si no elegimos país alguno, y queremos filtrar las atracciones solo por nombre, sin importar el país. Supongamos que queremos ver las atracciones comprendidas entre la "A" y la "T". Presionamos el botón List Attractions y vemos que el listado aparece vacío. ¿Por qué sucede esto?

Esto es porque en las cláusulas `where` del procedimiento, no contemplamos la posibilidad que alguno de los valores de las variables pudiera estar vacío, o sea que el usuario no ingrese ningún valor en ese filtro.

Para resolver esto, utilizamos las cláusulas `when`.

Al utilizar la cláusula `When` a continuación de la definición de la cláusula `Where`, estamos indicando que queremos aplicar una restricción a esta última. Veremos en futuros videos más en detalle su funcionamiento.

Por ejemplo, para el caso de nuestro primer `where`, agregamos al final `when not &NameFrom.isEmpty()`, o sea cuando la variable `NameFrom` no tenga valor vacío. Lo mismo hacemos para las otras dos condiciones.

Lo que estamos haciendo aquí, es indicando que queremos que se aplique la condición del `where` solamente cuando la variable tenga algún valor asignado, o sea, cuando no esté vacía. En caso contrario no se aplicará esta cláusula y se seguirá con la siguiente línea.

Ahora sí, guardamos, y volvemos a ejecutar con F5.


Seleccionamos el país Francia, y no aplicaremos filtros por nombre. Presionamos el botón List Attractions.

Nos aparece como esperábamos el listado con todas las atracciones de Francia.

Ahora probaremos dejando el país sin seleccionar, y queremos que se listen las atracciones comprendidas entre la "A" a la "G". Vemos que nos aparecen en el listado todas las atracciones que cumplan con esta condición, independientemente del país, que justamente es lo que buscábamos.

Como acabamos de ver, haciéndolo de este modo, podemos filtrar solo por país, solo por nombre, o por país y por nombre. Además de tener solo un procedimiento, y un único botón en nuestro web panel.

Enviamos todas las modificaciones a GxServer.



Para finalizar enviemos todo lo que hicimos a GeneXus Server.

En el siguiente video veremos otras formas de enviar y recibir parámetros, incluyendo los efectos de colocar un atributo en la regla Parm, en lugar de una variable.

GeneXus™

training.genexus.com
wiki.genexus.com
training.genexus.com/certifications